

Übung: Geheimcode

Ein einfacher Geheimcode funktioniert folgendermaßen: Der Sender einer Nachricht ersetzt in der Nachricht alle Buchstaben durch diejenigen Buchstaben, die sich vier Zeichen weiter hinten im Alphabet befinden, also A durch E, B durch F, C durch G usw. Bei den letzten vier Buchstaben (W-Z) kommt es dadurch zu einem „Überlauf“: W wird zu A, X zu B, Y zu C und Z zu D. Diese Nachricht wird dann verschickt.

Der Empfänger der Nachricht, muss die Verschlüsselung rückgängig machen, also alle Buchstaben durch die Buchstaben ersetzen, die sich vier Zeichen weiter vorne im Alphabet befinden: E wird wieder zu A, F zu B usw., und auch hier muss der „Überlauf“ berücksichtigt werden: A wird zu W, B zu X etc.

Die unverschlüsselte Nachricht nennen Kryptographen „Text“, die verschlüsselte „Cyphertext“.

Schreiben Sie zwei Sub-Prozeduren **Verschluesseln()** und **Entschluesseln()**, die sich sehr ähneln: Sie haben die folgende grundsätzliche Form:

```
Sub Verschluesseln()  
    Dim sText, sCyphertext As String  
    sText = InputBox ("Geben Sie die unverschlüsselte Nachricht ein.")  
    ...  
    sCyphertext = ...  
    MsgBox "Die verschlüsselte Nachricht lautet: " & sCyphertext  
End Sub
```

```
Sub Entschluesseln()  
    Dim sText, sCyphertext As String  
    sCyphertext = InputBox ("Geben Sie die verschlüsselte Nachricht ein.")  
    ...  
    sText = ...  
    MsgBox "Die entschlüsselte Nachricht lautet: " & sText  
End Sub
```

Ergänzen Sie in diesen beiden Sub-Prozeduren die fehlende Code-Teile, die (im ersten Fall) aus **sText** den verschlüsselten Text **sCyphertext** bzw. (im zweiten Fall) aus **sCyphertext** den ursprünglichen Klartext **sText** berechnen. Sie werden dabei buchstabenweise vorgehen müssen, dazu verwenden Sie eine Schleife der folgenden Form:

```
sCyphertext = ""           ' Ergebnis ist zunächst der leere String  
For i = 1 To Len (sText)  
    c = Mid(sText, i, 1)    ' c enthält das i-te Zeichen aus sText  
    z = ...                 ' z ist das Zeichen, das man erhält, wenn man c  
                           ' verschlüsselt  
    sCyphertext = sCyphertext + z  
Next i
```

Wenn Sie hier z. B. einfach **z = c** schreiben, kopiert diese Schleife einfach Zeichen für Zeichen den ersten String in den zweiten.

Verwenden Sie zum Umwandeln der einzelnen Zeichen die in der Vorlesung vorgestellten Funktionen **Asc()** und **Chr()**: Sie finden den Buchstaben, der vier Positionen weiter hinten im Alphabet steht, indem Sie den ASCII-Wert des Buchstabens berechnen, dann 4 addieren und aus dem erhaltenen Wert wieder ein Zeichen erzeugen. Überlegen Sie sich, wie Sie leicht erkennen können, ob es einen Überlauf gegeben hat, und was dann zu tun ist.

Zur Vereinfachung dürfen Sie davon ausgehen, dass alle Nachrichten ausschließlich Großbuchstaben enthalten; Sie müssen also nur mit den Zeichen A-Z (und nicht mit a-z) arbeiten.

Tipp: Lösen Sie zunächst das Teilproblem, einen einzelnen Buchstaben zu verschlüsseln (und wieder zu entschlüsseln). Wenn Sie wollen, können Sie für diese Aufgaben auch zwei Funktionen in der Form

```

Function VerschluesseleBuchstabe(c As String)
    ...
    VerschluesseleBuchstabe = ...
End Function

```

schreiben und aus den Sub-Prozeduren heraus aufrufen.

Lösung:

Option Explicit

```

Function VerschluesseleBuchstabe(ByVal c As String)
    Dim nWert As Integer
    nWert = Asc(c)
    nWert = nWert + 4           ' 4 addieren
    If nWert > 90 Then         ' 90 = "Z"
        nWert = nWert - 26
    End If
    VerschluesseleBuchstabe = Chr(nWert)
End Function

```

```

Function EntschluesseleBuchstabe(ByVal c As String)
    Dim nWert As Integer
    nWert = Asc(c)
    nWert = nWert - 4         ' 4 subtrahieren
    If nWert < 65 Then        ' 65 = "A"
        nWert = nWert + 26
    End If
    EntschluesseleBuchstabe = Chr(nWert)
End Function

```

```

Sub Verschluesseleln()
    Dim sText, sCyphertext, c, z As String
    Dim i As Integer
    sText = InputBox("Geben Sie die unverschlüsselte Nachricht ein.")
    sCyphertext = ""
    For i = 1 To Len(sText)
        c = Mid(sText, i, 1)           ' c enthält das i-te Zeichen aus sText
        z = VerschluesseleBuchstabe(c)
        sCyphertext = sCyphertext + z
    Next i
    MsgBox "Die verschlüsselte Nachricht lautet: " & sCyphertext
End Sub

```

```

Sub Entschluesseleln()
    Dim sText, sCyphertext, c, z As String
    Dim i As Integer
    sCyphertext = InputBox("Geben Sie die verschlüsselte Nachricht ein.")
    sText = ""
    For i = 1 To Len(sCyphertext)
        z = Mid(sCyphertext, i, 1)     ' z enthält das i-te Zeichen aus
                                        ' sCyphertext
        c = EntschluesseleBuchstabe(z)
        sText = sText + c
    Next i
    MsgBox "Die entschlüsselte Nachricht lautet: " & sText
End Sub

```